// Zombie Monster for KF Invasion gametype
class ZombieBossMini extends ZombieBoss;

function Died(Controller Killer, class<DamageType> damageType, vector HitLocation)
{
	Super(ZombieBossBase).Died(Killer,damageType,HitLocation);
}

state FireMissile
{
Ignores RangedAttack;

	function bool ShouldChargeFromDamage()
	{
		return false;
	}

	function BeginState()
	{
		Acceleration = vect(0,0,0);
	}

	function AnimEnd( int Channel )
	{
		local vector Start;
		local Rotator R;

		Start = GetBoneCoords('tip').Origin;

		if ( !SavedFireProperties.bInitialized )
		{
			SavedFireProperties.AmmoClass = MyAmmo.Class;
			SavedFireProperties.ProjectileClass = MyAmmo.ProjectileClass;
			SavedFireProperties.WarnTargetPct = 0.15;
			SavedFireProperties.MaxRange = 10000;
			SavedFireProperties.bTossed = False;
			SavedFireProperties.bTrySplash = False;
			SavedFireProperties.bLeadTarget = True;
			SavedFireProperties.bInstantHit = True;
			SavedFireProperties.bInitialized = true;
		}

		R = AdjustAim(SavedFireProperties,Start,100);
		PlaySound(RocketFireSound,SLOT_Interact,2.0,,TransientSoundRadius,,false);
		Spawn(Class'SelfGuidedProjectile',,,Start,R);
		Spawn(Class'SelfGuidedProjectile',,,Start,R);
		Spawn(Class'SelfGuidedProjectile',,,Start,R);
		bShotAnim = true;
		Acceleration = vect(0,0,0);
		SetAnimAction('FireEndMissile');
		HandleWaitForAnim('FireEndMissile');

		// Randomly send out a message about Patriarch shooting a rocket(5% chance)
		if ( FRand() < 0.05 && Controller.Enemy != none && PlayerController(Controller.Enemy.Controller) != none )
		{
			PlayerController(Controller.Enemy.Controller).Speech('AUTO', 10, "");
		}

		GoToState('');
	}
Begin:
	while ( true )
	{
		Acceleration = vect(0,0,0);
		Sleep(0.1);
	}
}

defaultproperties
{
     HealingLevels(0)=2800
     HealingLevels(1)=2600
     HealingLevels(2)=2400
     HealingAmount=500
     MGDamage=1.500000
     ClawMeleeDamageRange=25.000000
     ImpaleMeleeDamageRange=15.000000
     MeleeDamage=25
     ColOffset=(Z=32.500000)
     ColRadius=20.000000
     ColHeight=15.000000
     PlayerCountHealthScale=0.350000
     OnlineHeadshotOffset=(X=2.500000,Z=26.500000)
     OnlineHeadshotScale=1.500000
     bBoss=False
     ScoringValue=50
     HealthMax=4200.000000
     Health=4200
     HeadScale=0.800000
     MenuName="Mini Patriarch"
     DrawScale=0.525000
     PrePivot=(Z=-5.000000)
     CollisionHeight=25.000000
}
